On the Formalization of Lebesgue Integrals

Yasunari Shidama[1], Noburu Endou[2], and Pauline N. Kawamoto[1]

[1] Shinshu University, Nagano, Japan
[2] Gifu National College of Technology, Gifu, Japan

Abstract. In this paper, we report the progress of our work in the Mizar project of creating a library of various theorems relating to Lebesgue integrals. Concepts such as the integration of measurable functions are defined and topics on the linearity of integration operations, etc. are also included in this library.

1 Introduction

In this work, the authors focus on building a portion of the Mizar library dedicated to the area of functional analysis. The theory of Lebesgue integration, as with Riemann integration, is a classical as well as important foundation of analysis and provides a necessary part of the Mizar library in this area. Lebesgue integration is an important tool in probability and engineering fields and it is used widely in many applications. In the Mizar articles MESFUNC1 [7] to MESFUNC5 [10], we formalized the definitions for measurable functions, integration of simple functions, integration of measurable functions, as well as various theorems concerning properties such as linearity. Also, in the work up to MESFUNC5, integrated functions are assumed to take values of $\pm\infty$, but functions normally defined in Mizar are not designed to take $\pm\infty$ values in many cases. For this reason, in MESFUNC6 [12], we treat the definitions of integrability of general real valued functions which do not take values of $\pm\infty$ and their linearity, etc. We are also currently working on the formalization of various theorems related to these items as well as function spaces such as L^p space created by sets of integrable functions. This paper summarizes the work of the authors on the formalization of Lebesgue integration.

2 Outline of Formalization Work

2.1 Specification of Measurable Functions

The formalization of concepts concerning the foundations of σ fields and the theory of measurability was done in Mizar by Józef Białas in library articles MEASURE1 [1] to MEASURE4 [2]. In this work, the topic of measurability of sets is also addressed. However, the definition of measurability by Białas is of σ additive measure and is not always suitable for defining general measurable functions. For this reason, the authors redefined measurability of sets anew to apply only to σ fields. We compare these below.

ISBN 978-83-7431-128-1 ISSN 0860–150X

```
definition
  let X be set,
      S be SigmaField of X,
      M be sigma_Measure of S,
      A be set;
pred A is_measurable M means
:: MEASURE1:def 12
A in S;
end;

definition
  let X be set;
  let S be SigmaField of X;
  let A be set;
pred A is_measurable_on S means
:: MESFUNC1:def 11
A in S;
end;
```

The former is the definition by Białas and in regular mathematical terms `A is_measurable M` indicates an M-measurable set on measure space (X, S, M). The latter `A is_measurable_on S` on the other hand indicates a Borel set on measurable space (X, S). (In this sense, it might have been better to define the latter as `A is Borel`.) In MESFUNC1, the measurability of sets is redefined as mentioned above and the definition of measurable functions is constructed based on this. In the following MESFUNC2 [8], we discuss the sum, difference, etc. of measurable functions. For this, the treatment of algebraic operations and limits of real numbers including $\pm\infty$ (ExtREAL) becomes necessary so in EXTREAL1 [5], EXTREAL2 [6] we define the multiplication and division of ExtREAL and prove related theorems. A portion of the work is shown below.

```
definition
  let x,y be R_eal;
func x * y -> R_eal means
:: EXTREAL1:def 1
 (ex a,b being Real st (x = a & y = b & it = a * b)) or
 (((0. < x & y=+infty) or (0. < y & x=+infty) or (x < 0. & y=-infty)
 or (y < 0. & x = -infty)) & it = +infty) or
 (((x < 0. & y=+infty) or (y < 0. & x=+infty) or (0. < x & y=-infty)
 or (0. < y & x = -infty)) & it = -infty) or
 ((x = 0. or y = 0.) & it = 0.);
end;
```

The definition of measurable functions is formalized in MESFUNC1 as shown below by slightly generalizing the standard definition appearing in textbooks.

```
reserve X for non empty set;
reserve x for Element of X;
reserve f for PartFunc of X,ExtREAL;
reserve a for R_eal;

definition
  let X,f,a;
func less_dom(f,a) -> Subset of X means
:: MESFUNC1:def 12
x in it iff x in dom f & ex y being R_eal st y=f.x & y < a;

definition
  let X be non empty set;
  let S be SigmaField of X;
  let f be PartFunc of X,ExtREAL;
  let A be Element of S;
pred f is_measurable_on A means
:: MESFUNC1:def 17
for r being real number holds
    A /\ less_dom(f,R_EAL r) is_measurable_on S;
end;
```

Also, in the generalization here, although in standard textbooks measurable functions are taken to be functions defined on measurable sets (f is Function of A, ExtREAL), in this definition we use PartFunc to formulate the measurability of functions defined on sets that are not measurable. This is because some of the symbols will become complicated in the definition and we want to include cases of functions which are not defined on measurable functions.

The measurability of the sum, difference, and scalar product of measurable functions is as shown below.

```
reserve X for non empty set;
reserve f,g for PartFunc of X,ExtREAL;
reserve S for SigmaField of X;
reserve r for Real;
reserve A for Element of S;

theorem :: MESFUNC2:7
for f,g,A st f is_finite & g is_finite & f is_measurable_on A &
g is_measurable_on A holds f+g is_measurable_on A;

theorem :: MESFUNC2:13
   for f,g,A st f is_finite & g is_finite & f is_measurable_on A &
g is_measurable_on A & A c= dom g holds f-g is_measurable_on A;

theorem :: MESFUNC1:41
   for X,S,f,A,r st f is_measurable_on A & A c= dom f
```

```
holds r(#)f is_measurable_on A;
```

2.2 Formalization of Simple Functions

We defined the simple function in MESFUNC2 as follows.

```
registration
  let X be set;
  let S be SigmaField of X;
  cluster disjoint_valued FinSequence of S;
end;

definition let X be set;
  let S be SigmaField of X;
  mode Finite_Sep_Sequence of S is disjoint_valued FinSequence of S;
end;

definition
  let X be non empty set;
  let S be SigmaField of X;
  let f be PartFunc of X,ExtREAL;
pred f is_simple_func_in S means
:: MESFUNC2:def 5
 f is_finite &
 ex F being Finite_Sep_Sequence of S st
  (dom f = union rng F &
   for n being Nat,x,y being Element of X st
    n in dom F & x in F.n & y in F.n holds f.x = f.y);
end;
```

Here the meaning of F is `Finite_Sep_Sequence of S` is the values of F are a Finite Sequence of disjoint measurable sets. Therefore, to say that f is a simple function means that the domain of f can be divided into disjoint measurable sets of F and that for each $F.n$, f takes a constant value.

2.3 Integration of Simple Functions

In MESFUNC3 [11], we defined the integration of simple functions whose domains take non-empty, non-negative values as follows. The restriction on domains is resolved in MESFUNC5.

```
definition
  let X be non empty set;
  let S be SigmaField of X;
  let M be sigma_Measure of S;
  let f be PartFunc of X,ExtREAL;
    assume f is_simple_func_in S
```

```
        & dom f <> {}
        & for x be set st x in dom f holds 0. <= f.x;
  func integral(X,S,M,f) -> Element of ExtREAL means
  :: MESFUNC3:def 2
    ex F be Finite_Sep_Sequence of S,
        a, x be FinSequence of ExtREAL st F,a are_Re-presentation_of f
           & a.1 =0.
           & (for n be Nat st 2 <= n & n in dom a holds
             0. < a.n & a.n < +infty )
           & dom x = dom F
           & (for n be Nat st n in dom x holds x.n=a.n*(M*F).n)
           & it=Sum(x);
  end;
```

As is well known, simple functions can be expressed by dividing the domains of measurable sets (including measure 0) into a countless number of ways and producing an infinite number of expressions. We need to show that integration operations will always produce a unique answer regardless of the function expression used. To do this, we use FUNCTOR for the definition of integration and show its existence and uniqueness. Also, in MESFUNC4 [9] we show the linearity of integration of simple functions.

```
theorem :: MESFUNC4:5
for X be non empty set,
    S be SigmaField of X,
    M be sigma_Measure of S,
    f,g be PartFunc of X,ExtREAL
  st f is_simple_func_in S & dom f <> {}
   & (for x be set st x in dom f holds 0. <= f.x)
   & g is_simple_func_in S & dom g = dom f
   & (for x be set st x in dom g holds 0. <= g.x)
 holds
    f+g is_simple_func_in S & dom (f+g) <> {}
  & (for x be set st x in dom (f+g) holds 0. <= (f+g).x)
  & integral(X,S,M,f+g)=integral(X,S,M,f)+integral(X,S,M,g);

theorem :: MESFUNC4:6
for X be non empty set,
    S be SigmaField of X,
    M be sigma_Measure of S,
    f,g be PartFunc of X,ExtREAL, c be R_eal
  st f is_simple_func_in S
   & dom f <> {}
   & (for x be set st x in dom f holds 0. <= f.x)
   & 0. <= c & c < +infty
   & dom g = dom f
   & (for x be set st x in dom g holds g.x=c*f.x)
```

```
 holds
    integral(X,S,M,g)=c*integral(X,S,M,f);
```

2.4 Integration of Measurable Functions

To formalize the integration of measurable functions, we follow the method of standard textbooks and in MESFUNC5 we first show that for functions taking non-negative values it is expressed as the limit of a sequence of simple functions as:

```
definition
let X be non empty set,
    H be Functional_Sequence of X,ExtREAL,
    x be Element of X;
func H#x -> ExtREAL_sequence means
:: MESFUNC5:def 13
for n be Nat holds it.n = (H.n).x;
end;

theorem :: MESFUNC5:70
for X be non empty set, S be SigmaField of X,
    f be PartFunc of X,ExtREAL st
 (ex A be Element of S st A = dom f & f is_measurable_on A) &
 f is nonnegative
holds
  ex F be Functional_Sequence of X,ExtREAL st
   (for n be Nat holds F.n is_simple_func_in S & dom(F.n) = dom f) &
   (for n be Nat holds F.n is nonnegative) &
   (for n,m be Nat st n <=m holds
     for x be Element of X st x in dom f holds (F.n).x <= (F.m).x ) &
   (for x be Element of X st x in dom f holds
     (F#x) is convergent & lim(F#x) = f.x);
```

We define the limit of simple function integration as follows.

```
definition
let X be non empty set;
let S be SigmaField of X;
let M be sigma_Measure of S;
let f be PartFunc of X,ExtREAL;
  assume that
ex A be Element of S st A = dom f & f is_measurable_on A and
f is nonnegative;
func integral+(M,f) -> Element of ExtREAL means
:: MESFUNC5:def 15
ex F be Functional_Sequence of X,ExtREAL,
   K be ExtREAL_sequence st
```

```
    (for n be Nat holds F.n is_simple_func_in S & dom(F.n) = dom f) &
    (for n be Nat holds F.n is nonnegative) &
    (for n,m be Nat st n <=m holds
      for x be Element of X st x in dom f holds (F.n).x <= (F.m).x ) &
    (for x be Element of X st  x in dom f holds
      F#x is convergent & lim(F#x) = f.x) &
    (for n be Nat holds K.n=integral'(M,F.n)) &
    K is convergent &
    it=lim K;
end;
```

Here as well, to show that the limit does not depend on the simple function sequence selected, we use FUNCTOR, just as in the case of simple functions, to create the definition and prove the existence and uniqueness of the integration of a given non-negative measurable function.

Next, for functions which are not non-negative, we use the fact that they can be expressed using the non-negative function `max+f` and the non-positive function `max-f` and create the definition as follows.

```
definition
  let X be non empty set;
  let S be SigmaField of X;
  let M be sigma_Measure of S;
  let f be PartFunc of X,ExtREAL;
  func Integral(M,f) -> Element of ExtREAL equals
:: MESFUNC5:def 16
  integral+(M,max+f)-integral+(M,max-f);
end;
```

After we formalize the definition of integration for measurable functions, we use it to prepare a variety of theorems for Lebesgue integration and the formulation of function spaces such as L^p space. We show the representative theorems below.

– Theorems concerning the division of integration areas

```
theorem :: MESFUNC5:104
for X be non empty set,
    S be SigmaField of X,
    M be sigma_Measure of S,
    f be PartFunc of X,ExtREAL,
    A,B be Element of S st
 f is_integrable_on M & A misses B holds
  Integral(M,f|(A\/B)) = Integral(M,f|A) + Integral(M,f|B);

theorem :: MESFUNC5:105
for X be non empty set,
    S be SigmaField of X,
    M be sigma_Measure of S,
```

```
    f be PartFunc of X,ExtREAL,
    A,B be Element of S st
 f is_integrable_on M & B = (dom f)\A
holds
   f|A is_integrable_on M & Integral(M,f) = Integral(M,f|A)
                                           +Integral(M,f|B);
```

– Theorems on the integrability of absolute value functions of integrable functions

```
theorem :: MESFUNC5:106
for X be non empty set,
    S be SigmaField of X,
    M be sigma_Measure of S,
    f be PartFunc of X,ExtREAL st
 (ex A be Element of S st A = dom f & f is_measurable_on A )
   holds
    f is_integrable_on M iff |.f.| is_integrable_on M;

theorem :: MESFUNC5:107
for X be non empty set,
    S be SigmaField of X,
    M be sigma_Measure of S,
    f be PartFunc of X,ExtREAL st
 f is_integrable_on M holds
    |. Integral(M,f) .| <= Integral(M,|.f.|);

theorem :: MESFUNC5:108
for X be non empty set,
    S be SigmaField of X,
    M be sigma_Measure of S,
    f,g be PartFunc of X,ExtREAL st
 ( ex A be Element of S st A = dom f & f is_measurable_on A ) &
 dom f = dom g & g is_integrable_on M &
 ( for x be Element of X st x in dom f holds |.f.x .| <= g.x )
holds
   f is_integrable_on M & Integral(M,|.f.|) <= Integral(M,g);
```

– Theorem showing that integrable functions are sets with measure 0 that take $\pm\infty$ values

```
theorem :: MESFUNC5:111
for X be non empty set,
    S be SigmaField of X,
    M be sigma_Measure of S,
    f be PartFunc of X,ExtREAL st
f is_integrable_on M holds
  f"{+infty} in S & f"{-infty} in S & M.(f"{+infty})=0 &
```

```
                M.(f"{-infty})=0 &
f"{+infty} \/ f"{-infty} in S &
                M.(f"{+infty} \/ f"{-infty})=0;
```

– Theorems on the linearity of integration operations

```
theorem :: MESFUNC5:114
for X be non empty set,
    S be SigmaField of X,
    M be sigma_Measure of S,
    f,g be PartFunc of X,ExtREAL st
 f is_integrable_on M & g is_integrable_on M holds
   f+g is_integrable_on M;

theorem :: MESFUNC5:115
for X be non empty set,
    S be SigmaField of X,
    M be sigma_Measure of S,
    f,g be PartFunc of X,ExtREAL st
 f is_integrable_on M & g is_integrable_on M holds
  ex E be Element of S st
    E = dom f /\ dom g & Integral(M,f+g)=Integral(M,f|E)
                                        +Integral(M,g|E);

theorem :: MESFUNC5:116
for X be non empty set,
    S be SigmaField of X,
    M be sigma_Measure of S,
    f be PartFunc of X,ExtREAL,
    c be Real st
 f is_integrable_on M holds
  c(#)f is_integrable_on M &
  Integral(M,c(#)f) = R_EAL c * Integral(M,f);

definition
let X be non empty set;
let S be SigmaField of X;
let M be sigma_Measure of S;
let f be PartFunc of X,ExtREAL;
let B be Element of S;
func Integral_on(M,B,f) -> Element of ExtREAL equals
:: MESFUNC5:def 18
 Integral(M,f|B);
end;

theorem :: MESFUNC5:117
for X be non empty set,
```

```
      S be SigmaField of X,
      M be sigma_Measure of S,
      f,g be PartFunc of X,ExtREAL,
      B be Element of S st
  f is_integrable_on M & g is_integrable_on M & B c= dom(f+g)
 holds
    f+g is_integrable_on M &
    Integral_on(M,B,f+g) = Integral_on(M,B,f) + Integral_on(M,B,g);

 theorem :: MESFUNC5:118
 for X be non empty set,
      S be SigmaField of X,
      M be sigma_Measure of S,
      f be PartFunc of X,ExtREAL,
      c be Real,
      B be Element of S st
  f is_integrable_on M & f is_measurable_on B holds
   f|B is_integrable_on M &
   Integral_on(M,B,c(#)f) = R_EAL c * Integral_on(M,B,f);
```

The definitions and theorems above are for integration operations of functions which take values of real numbers including $\pm\infty$ (ExtREAL), but in spaces such as L^p, functions which take only finite real number values, not ExtREAL, are used. To conveniently connect them with the constructed library, we formulated a library on integration in MESFUNC6 for functions which take only finite real numbers as values.

3 Conclusion

We reported the current state of library construction concerning Lebesgue integrals. The fundamental areas including the definitions of measurability and integrability as well as the linearity of integration have been completed and the next step will be to use this work to formalize various related theorems as well as other types of function spaces such as L^p space. Also, since the current definition of integration uses measurability as a general σ additive measure, technically we cannot call this Lebesgue integration. For this reason, we must take a careful look at the Lebesgue measure defined in MEASURE7 [3]. Furthermore, we have not yet begun the work on integration of functions on direct product measure spaces and it will be necessary to begin this formalization as soon as possible.

The theory of Lebesgue integrals is considered a classical work and it contains delicate and rich contents. The authors of this paper share a deep interest in this area and would appreciate comments and collaboration in the construction efforts of this portion of the Mizar library.

Finally, the formalization of Lebesgue integration has been pursued thus far in the case where no topology has been introduced on set X. With regard to measures on topological spaces, the definition of Borel sets has been constructed in TOPGEN_4 [4]. Based on this, future work concerning the formalization of integration on topological spaces will become possible as a next step.

References

1. Józef Białas. The σ-additive Measure Theory. *Formalized Mathematics*, 2(**2**), pp.263-270, 1991.
2. Józef Białas. Properties of Caratheodor's Measure. *Formalized Mathematics*, 3(**1**), pp.67-70, 1992.
3. Józef Białas. The One-Dimensional Lebesgue Measure. *Formalized Mathematics*, 5(**2**), pp.253-258, 1996.
4. Adam Grabowski. On the Borel Families of Subsets of Topological Spaces. *Formalized Mathematics*, 13(**4**), pp.453-461, 2005.
5. Noboru Endou, Katsumi Wasaki, and Yasunari Shidama. Basic Properties of Extended Real Numbers. *Formalized Mathematics*, 9(**3**), pp.491-494, 2001.
6. Noboru Endou, Katsumi Wasaki, and Yasunari Shidama. Some Properties of Extended Real Numbers Operations: absolute value, min and max. *Formalized Mathematics*, 9(**3**), pp.511-516, 2001.
7. Noboru Endou, Katsumi Wasaki, and Yasunari Shidama. Definitions and Basic Properties of Measurable Functions. *Formalized Mathematics*, 9(**3**), pp.495-500, 2001.
8. Noboru Endou, Katsumi Wasaki, and Yasunari Shidama. Measurability of Extended Real Valued Functions. *Formalized Mathematics*, 9(**3**), pp.525-529, 2001.
9. Noboru Endou and Yasunari Shidama. Linearity of Lebesgue Integral of Simple Valued Function. *Formalized Mathematics*, 13(**4**), pp.463-466, 2005.
10. Noboru Endou and Yasunari Shidama. Integral of Measurable Function. *Formalized Mathematics*, 14(**2**), pp.53-70, 2006.
11. Yasunari Shidama and Noboru Endou. Lebesgue Integral of Simple Valued Function. *Formalized Mathematics*, 13(**1**), pp.67-72, 2005.
12. Yasunari Shidama and Noboru Endou. Integral of Real-Valued Measurable Function. *Formalized Mathematics*, 14(**4**), pp.143-152, 2006.